330 CS 538 Spring 2002^©

Sequential Composition

Expressions or statements are sequenced using “;”. Hence

val a = (1+2;3+4);
val a = 7 : int
(x:=1;!x);
val it = 1 : int

Iteration

while expr1 do expr2

implements iteration (and returns unit); Thus

(while false do 10);
val it = () : unit
while !x > 0 do x:= !x-1;
val it = () : unit
!x;
val it = 0 : int

331 CS 538 Spring 2002^©

Simple I/O

The function

 print;
 val it = fn : string -> unit

prints a string onto standard output. For example,

 print("Hello World\n");
 Hello World

The conversion routines

 Real.toString;
 val it = fn : real -> string
 Int.toString;
 val it = fn : int -> string
 Bool.toString;
 val it = fn : bool -> string

convert a value (real, int or bool) into astring. Unlike Java, the call must be explicit.

332 CS 538 Spring 2002^©

For example,

print(Int.toString(123));
123

Also available are

Real.fromString;
val it = fn : string -> real
option
Int.fromString;
val it = fn : string -> int
option
Bool.fromString;
val it = fn : string -> bool
option

which convert from astring to a

real or int or bool if possible.

(That’s why theoption type is used).

333 CS 538 Spring 2002^©

For example,

case (Int.fromString("123"))
 of
  SOME(i) => i | NONE => 0;
val it = 123 : int
case (Int.fromString(
      "One two three")) of
  SOME(i) => i | NONE => 0;
val it = 0 : int

334 CS 538 Spring 2002^©

Text I/O

ThestructureTextIO containsa wide variety of I/O types, values and functions.Youloadthesebyentering:

open TextIO;

Among the values loaded are

• type instream

This is the type that represents input text files.

• type outstream

This is the type that represents output text files.

• type vector = string

Makesvector a synonym for

 string.

• type elem = char

Makeselem a synonym forchar.

335 CS 538 Spring 2002^©

• val stdIn : instream
 val stdOut : outstream
 val stdErr : outstream

Predefined input and output streams.

• val openIn :
  string -> instream
 val openOut :
  string -> outstream

Open an input or output stream. For example,

 val out =
  openOut("/tmp/test1");
 val out = - : outstream

• val input :
   instream -> vector

Read a line of input into astring (vector is defined as equivalent to string). For example (user input is in red):

336 CS 538 Spring 2002^©

  val s = input(stdIn);
  Hello!
  val s = "Hello!\n" : vector

• val inputN :
  instream * int -> vector

Readthe nextN inputcharactersinto a string. For example,

  val t = inputN(stdIn,3);
  abcde
  val t = "abc" : vector

• val inputAll :
  instream -> vector

Read the rest of the input file into a string (with newlines separating lines). For example,

 val u = inputAll(stdIn);
 Four score and
 seven years ago ...
 val u = "Four score and\nseven
 years ago ...\n" : vector

337 CS 538 Spring 2002^©

• val endOfStream :
  instream -> bool

Are we at the end of this input stream?

• val output :
  outstream * vector -> unit

Output astring on the specified output stream. For example,

output(stdOut,
 "That’s all folks!\n");
That’s all folks!

338 CS 538 Spring 2002^©

String Operations

ML provides a wide variety of string manipulation routines. Included are:

• The string concatenation operator, ^
 "abc" ^ "def" = "abcdef"
• The standard 6 relational operators:

< > <= >= = <>

• The string size operator:
 val size : string -> int
 size ("abcd");
 val it = 4 : int

• The string subscripting operator

(indexing from 0):

val sub =
 fn : string * int -> char
sub("abcde",2);
val it = #"c" : char

339 CS 538 Spring 2002^©

• The substring function
 val substring :
 string * int * int -> string

This function is called as

substring(string,start,len)
start is the starting position,

counting from 0.

len is the length of the desired substring. For example,

 substring("abcdefghij",3,4)
 val it = "defg" : string

• Concatenationofa listofstringsinto

a single string:

concat :
 string list -> string

For example,

concat ["What’s"," up","?"];
val it = "What’s up?" : string

340 CS 538 Spring 2002^©

• Convert a character into a string:
 str : char -> string

For example,

  str(#"x");
   val it = "x" : string

• “Explode” a string into a list of

characters:

explode : string -> char list

For example,

 explode("abcde");
 val it =
 [#"a",#"b",#"c",#"d",#"e"] :
 char list

• “Implode” a list of characters into a

string.

implode : char list -> string

For example,

implode
[#"a",#"b",#"c",#"d",#"e"];
val it = "abcde" : string

341 CS 538 Spring 2002^©

Structures and Signatures

In C++ and Java you can group variableandfunctiondefinitionsinto classes. In Java you can also group classes into packages. InMLyoucangroupvalue,exception and function definitions into structures. You can then import selected definitions from the structure (using the notationstructure.name) or you canopen the structure, thereby importing all the definitions within the structure. (Examplesusedinthissectionmaybe found at

~cs538-1/public/sml/struct.sml)

342 CS 538 Spring 2002^©

The general form of a structure definition is

structure name =
struct
  val, exception and
  fun definitions
end

For example,

structure Mapping =
struct
  exception NotFound;
  val create = [];
  fun lookup(key,[]) =
      raise NotFound
   |  lookup(key,
            (key1,value1)::rest) =
      if key = key1
      then value1
      else lookup(key,rest);

343 CS 538 Spring 2002^©

  fun insert(key,value,[]) =
                 [(key,value)]
   |  insert(key,value,
         (key1,value1)::rest) =
      if key = key1
      then (key,value)::rest
      else (key1,value1)::
            insert(key,value,rest);
end;

We can access members of this structure asMapping.name. Thus

Mapping.insert(538,"languages",[]);
val it = [(538,"languages")] :
(int * string) list
open Mapping;
exception NotFound
val create : 'a list
val insert : ''a * 'b * (''a * 'b)
 list -> (''a * 'b) list
val lookup : ''a * (''a * 'b)
 list -> 'b

344 CS 538 Spring 2002^©

Signatures

Eachstructurehasa signature, which is it type. For example,Mapping’s signature is

structure Mapping :
  sig
    exception NotFound
    val create : 'a list
    val insert : ''a * 'b *
       (''a * 'b) list ->
       (''a * 'b) list
    val lookup : ''a *
        (''a * 'b) list -> 'b
  end

345 CS 538 Spring 2002^©

You can define a signature as

signature name = sig
  type definitions for values,
   functions and exceptions
end

For example,

signature Str2IntMapping =
sig
  exception NotFound;
  val lookup:
    string * (string*int) list
      -> int;
end;

346 CS 538 Spring 2002^©

Signatures can be used to

• Restrict the type of a value or function in a structure.

• Hide selected definitions that appear in a structure For example

structure Str2IntMap :
  Str2IntMapping = Mapping;

definesa newstructure,Str2IntMap, createdbyrestrictingMapping to the

Str2IntMapping signature. When

we do this we get

open  Str2IntMap;
  exception NotFound
  val lookup : string *
   (string * int) list -> int

Onlylookup and NotFound are created,andlookup islimitedto keys that are strings.

347 CS 538 Spring 2002^©

Extending ML’s Polymorphism

In languages like C++ and Java we mustusetypeslikevoid* or Object to simulate the polymorphism that MLprovides.InMLwheneverpossible a general type (a polytype) is used rather than a fixed type. Thus in

 fun len([]) = 0
  | len(a::b) = 1 + len(b);

we get a type of

 'a list -> int

because this is the most general type possible that is consistent withlen’s definition. Is this form of polymorphism general enoughto capturethegeneralideaof making program definitions as typeindependent as possible?

348 CS 538 Spring 2002^©

It isn’t, and to see why consider the following ML definition of a merge sort. A merge sort operates by first splitting a list into two equal length sublists. The following function does this:

fun split [] = ([],[])
 |  split [a] = ([a],[])
 |  split (a::b::rest) =
    let val (left,right) =
      split(rest) in
       (a::left, b::right)
    end;

After the input list is split into two halves, eachhalfisrecursivelysorted, then the sorted halves are merged together into a single list. The following ML function merges two sorted lists into one:

349 CS 538 Spring 2002^©

fun merge([],[]) = []
 |  merge([],hd::tl) = hd::tl
 |  merge(hd::tl,[]) = hd::tl
 |  merge(hd::tl,h::t) =
    if hd <= h
    then hd::merge(tl,h::t)
    else h::merge(hd::tl,t)

With these two subroutines, a definition of a sort is easy:

fun sort [] = []
 |  sort([a]) = [a]
 |  sort(a::b::rest) =
    let val (left,right) =
       split(a::b::rest) in
      merge(sort(left),
            sort(right))
    end;

350 CS 538 Spring 2002^©

This definition looks very general—it should work for a list of any type. Unfortunately, when ML types the functions we get a surprise:

val split = fn : 'a list ->
 'a list * 'a list
val merge = fn : int list *
 int list -> int list
val sort = fn :
 int list -> int list
split ispolymorphic,but merge and
sort are limited to integer lists!

Wheredidthisrestrictioncomefrom?

351 CS 538 Spring 2002^©

The problem is that we did a comparison inmerge using the<= operator, and ML typed this as an integer comparison. We can make our definition of sort moregeneralbyaddinga comparison function,le(a,b) as a parameter to

merge and sort. If we curry this

parameter we may be able to hide it from end users. Our updated definitions are:

fun merge(le,[],[]) = []
 |  merge(le,[],hd::tl) = hd::tl
 |  merge(le,hd::tl,[]) = hd::tl
 |  merge(le,hd::tl,h::t) =
    if le(hd,h)
    then hd::merge(le,tl,h::t)
    else h::merge(le,hd::tl,t)

352 CS 538 Spring 2002^©

fun sort le [] = []
 |  sort le [a] = [a]
 |  sort le (a::b::rest) =
     let val (left,right) =
       split(a::b::rest) in
         merge(le, sort le left,
                   sort le right)
     end;

Now the types ofmerge and sort are:

val merge = fn :
 ('a * 'a -> bool) *
  'a list * 'a list -> 'a list
val sort = fn : ('a * 'a -> bool)
      -> 'a list -> 'a list

We can now “customize”sort by choosing a particular definition for the le parameter:

fun le(a,b) = a <= b;
val le = fn : int * int -> bool

353 CS 538 Spring 2002^©

fun intsort L = sort le L;
val intsort =
  fn : int list -> int list
intsort(
 [4,9,0,2,111,~22,8,~123]);
val it = [~123,~22,0,2,4,8,9,111]
: int list
fun strle(a:string,b) =
   a <= b;
val strle =
  fn : string * string -> bool
fun strsort L = sort strle L;
val strsort =
 fn : string list -> string list
strsort(
  ["aac","aaa","ABC","123"]);
val it =
["123","ABC","aaa","aac"] :
string list